#ifndef EXAMPLE_GLSL_H
#define EXAMPLE_GLSL_H
static const char* EXAMPLE_GLSL1 = R"(
 vec3 palette(float d){
         return mix(vec3(0.2,0.7,0.9),vec3(1.,0.,1.),d);
 }

 vec2 rotate(vec2 p,float a){
     float c = cos(a);
     float s = sin(a);
     return p*mat2(c,s,-s,c);
 }

 float map(vec3 p){
     for( int i = 0; i<8; ++i){
         float t = iTime*0.2;
         p.xz =rotate(p.xz,t);
         p.xy =rotate(p.xy,t*1.89);
         p.xz = abs(p.xz);
         p.xz-=.5;
     }
     return dot(sign(p),p)/5.;
 }

 vec4 rm (vec3 ro, vec3 rd){
     float t = 0.;
     vec3 col = vec3(0.);
     float d;
     for(float i =0.; i<64.; i++){
         vec3 p = ro + rd*t;
         d = map(p)*.5;
         if(d<0.02){
             break;
         }
         if(d>100.){
             break;
         }
         //col+=vec3(0.6,0.8,0.8)/(400.*(d));
         col+=palette(length(p)*.1)/(400.*(d));
         t+=d;
     }
     return vec4(col,1./(d*100.));
 }
 void mainImage( out vec4 fragColor, in vec2 fragCoord )
 {
     vec2 uv = (fragCoord-(iResolution.xy/2.))/iResolution.x;
     vec3 ro = vec3(0.,0.,-50.);
     ro.xz = rotate(ro.xz,iTime);
     vec3 cf = normalize(-ro);
     vec3 cs = normalize(cross(cf,vec3(0.,1.,0.)));
     vec3 cu = normalize(cross(cf,cs));
     vec3 uuv = ro+cf*3. + uv.x*cs + uv.y*cu;
     vec3 rd = normalize(uuv-ro);
     vec4 col = rm(ro,rd);
     fragColor = col;
 }
)";
static const char* EXAMPLE_GLSL2 = R"(
/*
 * "Seascape" by Alexander Alekseev aka TDM - 2014
 * License Creative Commons Attribution-NonCommercial-ShareAlike 3.0 Unported License.
 * Contact: tdmaav@gmail.com
 */

const int NUM_STEPS = 8;
const float PI	 	= 3.141592;
const float EPSILON	= 1e-3;
#define EPSILON_NRM (0.1 / iResolution.x)
#define AA

// sea
const int ITER_GEOMETRY = 3;
const int ITER_FRAGMENT = 5;
const float SEA_HEIGHT = 0.6;
const float SEA_CHOPPY = 4.0;
const float SEA_SPEED = 0.8;
const float SEA_FREQ = 0.16;
const vec3 SEA_BASE = vec3(0.0,0.09,0.18);
const vec3 SEA_WATER_COLOR = vec3(0.8,0.9,0.6)*0.6;
#define SEA_TIME (1.0 + iTime * SEA_SPEED)
const mat2 octave_m = mat2(1.6,1.2,-1.2,1.6);

// math
mat3 fromEuler(vec3 ang) {
    vec2 a1 = vec2(sin(ang.x),cos(ang.x));
    vec2 a2 = vec2(sin(ang.y),cos(ang.y));
    vec2 a3 = vec2(sin(ang.z),cos(ang.z));
    mat3 m;
    m[0] = vec3(a1.y*a3.y+a1.x*a2.x*a3.x,a1.y*a2.x*a3.x+a3.y*a1.x,-a2.y*a3.x);
    m[1] = vec3(-a2.y*a1.x,a1.y*a2.y,a2.x);
    m[2] = vec3(a3.y*a1.x*a2.x+a1.y*a3.x,a1.x*a3.x-a1.y*a3.y*a2.x,a2.y*a3.y);
    return m;
}
float hash( vec2 p ) {
    float h = dot(p,vec2(127.1,311.7));
    return fract(sin(h)*43758.5453123);
}
float noise( in vec2 p ) {
    vec2 i = floor( p );
    vec2 f = fract( p );
    vec2 u = f*f*(3.0-2.0*f);
    return -1.0+2.0*mix( mix( hash( i + vec2(0.0,0.0) ),
                     hash( i + vec2(1.0,0.0) ), u.x),
                mix( hash( i + vec2(0.0,1.0) ),
                     hash( i + vec2(1.0,1.0) ), u.x), u.y);
}

// lighting
float diffuse(vec3 n,vec3 l,float p) {
    return pow(dot(n,l) * 0.4 + 0.6,p);
}
float specular(vec3 n,vec3 l,vec3 e,float s) {
    float nrm = (s + 8.0) / (PI * 8.0);
    return pow(max(dot(reflect(e,n),l),0.0),s) * nrm;
}

// sky
vec3 getSkyColor(vec3 e) {
    e.y = (max(e.y,0.0)*0.8+0.2)*0.8;
    return vec3(pow(1.0-e.y,2.0), 1.0-e.y, 0.6+(1.0-e.y)*0.4) * 1.1;
}

// sea
float sea_octave(vec2 uv, float choppy) {
    uv += noise(uv);
    vec2 wv = 1.0-abs(sin(uv));
    vec2 swv = abs(cos(uv));
    wv = mix(wv,swv,wv);
    return pow(1.0-pow(wv.x * wv.y,0.65),choppy);
}

float map(vec3 p) {
    float freq = SEA_FREQ;
    float amp = SEA_HEIGHT;
    float choppy = SEA_CHOPPY;
    vec2 uv = p.xz; uv.x *= 0.75;

    float d, h = 0.0;
    for(int i = 0; i < ITER_GEOMETRY; i++) {
        d = sea_octave((uv+SEA_TIME)*freq,choppy);
        d += sea_octave((uv-SEA_TIME)*freq,choppy);
        h += d * amp;
        uv *= octave_m; freq *= 1.9; amp *= 0.22;
        choppy = mix(choppy,1.0,0.2);
    }
    return p.y - h;
}

float map_detailed(vec3 p) {
    float freq = SEA_FREQ;
    float amp = SEA_HEIGHT;
    float choppy = SEA_CHOPPY;
    vec2 uv = p.xz; uv.x *= 0.75;

    float d, h = 0.0;
    for(int i = 0; i < ITER_FRAGMENT; i++) {
        d = sea_octave((uv+SEA_TIME)*freq,choppy);
        d += sea_octave((uv-SEA_TIME)*freq,choppy);
        h += d * amp;
        uv *= octave_m; freq *= 1.9; amp *= 0.22;
        choppy = mix(choppy,1.0,0.2);
    }
    return p.y - h;
}

vec3 getSeaColor(vec3 p, vec3 n, vec3 l, vec3 eye, vec3 dist) {
    float fresnel = clamp(1.0 - dot(n,-eye), 0.0, 1.0);
    fresnel = pow(fresnel,3.0) * 0.5;

    vec3 reflected = getSkyColor(reflect(eye,n));
    vec3 refracted = SEA_BASE + diffuse(n,l,80.0) * SEA_WATER_COLOR * 0.12;

    vec3 color = mix(refracted,reflected,fresnel);

    float atten = max(1.0 - dot(dist,dist) * 0.001, 0.0);
    color += SEA_WATER_COLOR * (p.y - SEA_HEIGHT) * 0.18 * atten;

    color += vec3(specular(n,l,eye,60.0));

    return color;
}

// tracing
vec3 getNormal(vec3 p, float eps) {
    vec3 n;
    n.y = map_detailed(p);
    n.x = map_detailed(vec3(p.x+eps,p.y,p.z)) - n.y;
    n.z = map_detailed(vec3(p.x,p.y,p.z+eps)) - n.y;
    n.y = eps;
    return normalize(n);
}

float heightMapTracing(vec3 ori, vec3 dir, out vec3 p) {
    float tm = 0.0;
    float tx = 1000.0;
    float hx = map(ori + dir * tx);
    if(hx > 0.0) {
        p = ori + dir * tx;
        return tx;
    }
    float hm = map(ori + dir * tm);
    float tmid = 0.0;
    for(int i = 0; i < NUM_STEPS; i++) {
        tmid = mix(tm,tx, hm/(hm-hx));
        p = ori + dir * tmid;
        float hmid = map(p);
        if(hmid < 0.0) {
            tx = tmid;
            hx = hmid;
        } else {
            tm = tmid;
            hm = hmid;
        }
    }
    return tmid;
}

vec3 getPixel(in vec2 coord, float time) {
    vec2 uv = coord / iResolution.xy;
    uv = uv * 2.0 - 1.0;
    uv.x *= iResolution.x / iResolution.y;

    // ray
    vec3 ang = vec3(sin(time*3.0)*0.1,sin(time)*0.2+0.3,time);
    vec3 ori = vec3(0.0,3.5,time*5.0);
    vec3 dir = normalize(vec3(uv.xy,-2.0)); dir.z += length(uv) * 0.14;
    dir = normalize(dir) * fromEuler(ang);

    // tracing
    vec3 p;
    heightMapTracing(ori,dir,p);
    vec3 dist = p - ori;
    vec3 n = getNormal(p, dot(dist,dist) * EPSILON_NRM);
    vec3 light = normalize(vec3(0.0,1.0,0.8));

    // color
    return mix(
        getSkyColor(dir),
        getSeaColor(p,n,light,dir,dist),
        pow(smoothstep(0.0,-0.02,dir.y),0.2));
}

// main
void mainImage( out vec4 fragColor, in vec2 fragCoord ) {
    float time = iTime * 0.3 + iMouse.x*0.01;

#ifdef AA
    vec3 color = vec3(0.0);
    for(int i = -1; i <= 1; i++) {
        for(int j = -1; j <= 1; j++) {
            vec2 uv = fragCoord+vec2(i,j)/3.0;
            color += getPixel(uv, time);
        }
    }
    color /= 9.0;
#else
    vec3 color = getPixel(fragCoord, time);
#endif

    // post
    fragColor = vec4(pow(color,vec3(0.65)), 1.0);
}
)";

static const char* EXAMPLE_GLSL3 = R"(
    /*

    Greetings
    All Revision participants
    All Live Shader coders

    In particular Gaz and Kamoshika that I heavly based the code on their work :D

    */

    // I CERTIFY ITS NOT A BOT

    mat2 rot(float a){float c=cos(a),s=sin(a);return mat2(c,-s,s,c);}
    float diam(vec2 p,float s){
       p = abs(p);
       return (p.x+p.y-s)*inversesqrt(3.);

    }
    float smin(float a,float b,float r){
        float k = max(0.,r-abs(a-b));
      return min(a,b) -k*k*.25/r;

    }

    void mainImage( out vec4 fragColor, in vec2 fragCoord )
    {
       float bpm = (iTime*60./130.*2.);
        // Normalized pixel coordinates (from 0 to 1)
        vec2 uv = (fragCoord-.5*iResolution.xy)/iResolution.y;

        bpm = floor(bpm)+pow(fract(bpm),.5);

        vec3 col = vec3(.1);
        vec3 p,d = normalize(vec3(uv,1.));

        for(float i=0.,g=0.,e;i++<99.;){

          p = d*g;
          p.z -=5.;

          vec3 gp = p;
          gp.xy *=rot(gp.z*.1);
          gp.y =-abs(gp.y);
          gp.y +=1.;


          float dd,c=20./3.141592;

          gp.xz = vec2(log(dd=length(gp.xz)),atan(p.x,p.z))*c;
                                                // Here I struggle during live
                                                // as I was doing p.y (which is to do torus)
                                                // Rather than atan(p.x,p.y) to have proper log polar
          gp.y/=dd/=c;
          gp.y +=sin(gp.x)*.5;
          gp.xz = fract(gp.xz+iTime)-.5;

          for(float j=0.;j<4.;j++){
            gp.xzy = abs(gp.xzy)-vec3(.1,.01,.1);
             gp.xz *=rot(-.785);
          }
          float ha_grid = dd*.8*min(diam(gp.xy,.01),diam(gp.zy,.01));
               // You're a variable Harry


          float f = ha_grid;

          float blob = length(p)-.5;
          float gy = dot(sin(p*4.),cos(p.zxy*2.))*.1;
          for(float j=0.;j<16.;j++){
               vec3 off = vec3(cos(j),tan(bpm+j),sin(j*3.33))+gy;
                blob = smin(blob,length(p-off)-.125,.25);
          }

          f= smin(f,blob,.5);
          g+=e=max(.001,f);;
          col+= mix(vec3(1.,.2,sin(p.z+bpm)*.5+.5),vec3(.5,sin(p.z)*.5+.5,.9),fract(2.*i*i*e))*.25/exp(i*i*e);

       }
        // Output to screen
        fragColor = vec4(col,1.0);
    }
)";
#endif // EXAMPLE_GLSL_H
